Computer Physics Communications 184 (2013) 1581–1591

Contents lists available at SciVerse ScienceDirect

Computer Physics Communications

journal homepage: www.elsevier.com/locate/cpc

Efficient computation of Wigner–Eisenbud functions☆

Bahaaudin M. Raffah[a,b,*], Paul C. Abbott[a]

[a] School of Physics, M013, The University of Western Australia, 35 Stirling Highway, Crawley, WA 6009, Australia
[b] Physics department, Faculty of Science, King Abdulaziz University, P O Box 80203, Jeddah, Saudi Arabia

ARTICLE INFO

Article history:
Received 7 August 2012
Accepted 30 December 2012
Available online 8 February 2013

Keywords:
Wigner–Eisenbud functions
Discrete cosine transform (DCT)
Semiconductor heterostructures
Quantum transport
Cylindrical nanowires

ABSTRACT

The *R*-matrix method, introduced by Wigner and Eisenbud (1947) [1], has been applied to a broad range of electron transport problems in nanoscale quantum devices. With the rapid increase in the development and modeling of nanodevices, efficient, accurate, and general computation of Wigner–Eisenbud functions is required. This paper presents the *Mathematica* package `WignerEisenbud`, which uses the Fourier discrete cosine transform to compute the Wigner–Eisenbud functions in dimensionless units for an arbitrary potential in one dimension, and two dimensions in cylindrical coordinates.

Program summary

Program title: WignerEisenbud

Catalogue identifier: AEOU_v1_0

Program summary URL: http://cpc.cs.qub.ac.uk/summaries/AEOU_v1_0.html

Program obtainable from: CPC Program Library, Queen's University, Belfast, N. Ireland

Licensing provisions: Standard CPC licence, http://cpc.cs.qub.ac.uk/licence/licence.html

Distribution format: tar.gz

Programming language: Mathematica

Operating system: Any platform supporting *Mathematica* 7.0 and above

Keywords: Wigner-Eisenbud functions, discrete cosine transform (DCT), cylindrical nanowires

Classification: 7.3, 7.9, 4.6, 5

Nature of problem:
Computing the 1D and 2D Wigner–Eisenbud functions for arbitrary potentials using the DCT.

Solution method:
The R-matrix method is applied to the physical problem. Separation of variables is used for eigenfunction expansion of the 2D Wigner–Eisenbud functions. Eigenfunction computation is performed using the DCT to convert the Schrödinger equation with Neumann boundary conditions to a generalized matrix eigenproblem.

Limitations: Restricted to uniform (rectangular grid) sampling of the potential. In 1D the number of sample points, n, results in matrix computations involving $n \times n$ matrices.

Unusual features:
Eigenfunction expansion using the DCT is fast and accurate. Users can specify scattering potentials using functions, or interactively using mouse input. Use of dimensionless units permits application to a wide range of physical systems, not restricted to nanoscale quantum devices.

Running time: Case dependent.

☆ This paper and its associated computer program are available via the Computer Physics Communication homepage on ScienceDirect (http://www.sciencedirect.com/science/journal/00104655).

* Corresponding author at: School of Physics, M013, The University of Western Australia, 35 Stirling Highway, Crawley, WA 6009, Australia. Tel.: +61 415185915.
E-mail addresses: 10571115@student.uwa.edu.au, Braffah2003@yahoo.com (B.M. Raffah), paul.c.abbott@uwa.edu.au (P.C. Abbott).

0010-4655/$ – see front matter © 2013 Elsevier B.V. All rights reserved.
doi:10.1016/j.cpc.2012.12.027

1. Introduction

Computing the scattering wavefunction was a difficult problem in the early days of quantum mechanics. In the late 1940s, Wigner and Eisenbud [1] established a new technique for treating nuclear reaction scattering problems, now known as the *R*-matrix method. Applications to atomic and molecular processes, such as electron-molecule scattering, are discussed by Burke et al. [2], and reviewed in Burke and Berrington [3]. More recently, this method has been applied to a broad range of transport problems in nanoscale quantum devices [4–8]. The *R*-matrix method has wide application to transport phenomena [9], where the method has been applied to calculate the transmission of electromagnetic modes in microcavities [10].

Of course, many different methods have been applied to transport problems [11–15], but we focus on the use of Wigner–Eisenbud functions (WEFs). The Wigner–Eisenbud method is preferable for several reasons: (1) this approach is not affected by sudden scattering potential changes, nor even when the geometry of the system changes [16]; (2) numerically, the Wigner–Eisenbud method is highly efficient compared with other approaches; and (3) it is suitable for higher-dimensional nanostructures, with complex geometry and general nonseparable scattering potentials [17]. The most important feature of the Wigner–Eisenbud problem is that the number of zeros of the WEF is the same as in the self-consistent procedure [5, Section 3.3.1].

Applications require code that evaluate WEFs quickly and accurately for a wide range of problems, and for arbitrary potentials. Furthermore, as research into nanodevices expands [18,19], there will be an increase in applications requiring WEFs for a variety of different problems.

The Wigner–Eisenbud problem can be defined as the eigenvalue problem of the Hamiltonian operator for the closed equivalent of the open quantum system under consideration [17]. Using WEFs in the scattering region permits matching the wavefunction in the asymptotic region to that inside the scattering region [18,20]. The WEFs satisfy the Schrödinger equation, but with Neumann boundary conditions [21]; separation of variables is used to obtain WEFs in two dimensions.

This paper presents the *Mathematica* package `WignerEisenbud`, which computes WEFs in one dimension and two dimensions in cylindrical coordinates. Section 2 describes the theory in one and two dimensions. Use of dimensionless units permits application of the Wigner–Eisenbud method to a wide range of physical systems, not restricted to nanoscale quantum devices. In Section 3, computation of the WEF is performed using the Fourier discrete cosine transform (DCT). Examples of various potentials are then presented with visualizations of the corresponding WEF.

Further applications of the functionality provided by `WignerEisenbud` to transport computations in cylindrical quantum nanowires are presented in Raffah and Abbott [22].

2. Theory

2.1. 1D Wigner–Eisenbud functions

The 1D WEF, $X_l(z)$, satisfies the Schrödinger equation

$$\left(-\frac{\hbar^2}{2\mu}\frac{\partial^2}{\partial z^2}+\mathcal{V}(z)\right)X_l(z)=\mathcal{E}_l X_l(z), \tag{1}$$

where μ is the effective electron mass, $\mathcal{V}(z)$ is the potential over $-d\le z\le d$, and the eigenvalues, $\mathcal{E}_l$, are determined by applying the Neumann boundary conditions [18,20],

$$\chi_l'(\pm d)=0. \tag{2}$$

Without loss of generality, after scaling lengths by d and energies by $\hbar^2/\left(2\mu d^2\right)$, the *dimensionless* WEFs $\chi_l(z)$ arise as solutions to the *scaled* Schrödinger equation

$$-\chi_l''(z)+V(z)\chi_l(z)=E_l\chi_l(z), \tag{3}$$

over $[-1,1]$ with Neumann boundary conditions $\chi_l'(\pm1)=0$. Solutions to the physical problem are then obtained by the following substitutions:

$$X_l(zd)=\chi_l(z),\qquad \mathcal{V}(zd)=\frac{\hbar^2}{2\mu d^2}V(z),\qquad \mathcal{E}_l=\frac{\hbar^2}{2\mu d^2}E_l. \tag{4}$$

To compute $\chi_l(z)$ we write

$$\chi_l(z)=\sum_{n=0}^{\infty}c_n^{(l)}\varphi_n(z), \tag{5}$$

where $\varphi_a(z)$ is the following orthonormal trigonometric basis set (corresponding to the DCT of type 3, DCT-III),

$$\varphi_a(z)=\begin{cases}\dfrac{1}{\sqrt{2}} & a=0;\\ (-1)^{(a+1)/2}\sin\left(a\dfrac{\pi}{2}z\right) & \text{for odd } a;\\ (-1)^{a/2}\cos\left(a\dfrac{\pi}{2}z\right) & \text{for even } a.\end{cases} \tag{6}$$



Fig. 1. Geometry for a cylindrical nanowire.

$\varphi_a(z)$ clearly satisfies the Neumann boundary conditions $\varphi_a'(\pm 1) = 0$. Substituting Eq. (6) into Eq. (3), and multiplying both sides by $\varphi_b(z)$, yields the matrix version of the scaled Schrödinger equation (3). Evaluation of the kinetic energy matrix T is immediate,

$$\int_{-1}^{1} \varphi_a'(z)\varphi_b'(z)dz = \left(\frac{\pi a}{2}\right)^2 \delta_{a,b}, \tag{7}$$

leading to the following matrix problem,

$$\left(\frac{\pi a}{2}\right)^2 \delta_{a,b} + \langle a\,|V|\,b\rangle = E\,\langle a|b\rangle\,, \tag{8}$$

where we have introduced the potential energy matrix element

$$\langle a\,|V|\,b\rangle = \int_{-1}^{1} \varphi_a(z)V(z)\varphi_b(z)dz. \tag{9}$$

Efficient computation of the matrix elements (9) using the DCT is described in Section 3.

2.2. 2D problem

In this section, we consider a 2D cylindrical geometry (Fig. 1). The 2D Schrödinger equation in cylindrical coordinates reads [16]:

$$\left\{-\frac{\hbar^2}{2\mu}\left(\frac{\partial^2}{\partial r^2} + \frac{1}{r}\frac{\partial}{\partial r} - \frac{m^2}{r^2} + \frac{\partial^2}{\partial z^2}\right) + \mathcal{V}(r,z)\right\} X(r,z) = \mathcal{E}X(r,z), \tag{10}$$

where the Wigner–Eisenbud index l has been suppressed. Without loss of generality, after scaling lengths ($r \to rR$, $z \to zd$) and energies $\left(\mathcal{E} \to \hbar^2 / \left(2\mu d^2\right)\mathcal{E}\right)$, and introducing the *dimensionless scale factor* $s = d/R$ (see Fig. 1), Eq. (10) becomes

$$H\chi(r,z) = \left(s^2 T_r^{(m)} + T_z + V(r,z)\right)\chi(r,z) = E\chi(r,z), \tag{11}$$

where $T_r^{(m)}$ and T_z are the radial and axial parts, respectively, of the kinetic energy operator

$$T_r^{(m)} = -\left(\frac{\partial^2}{\partial r^2} + \frac{1}{r}\frac{\partial}{\partial r} - \frac{m^2}{r^2}\right), \qquad T_z = -\frac{\partial^2}{\partial z^2}. \tag{12}$$

Applying the boundary conditions

$$\chi(1,z) = 0, \qquad \chi^{(0,1)}(r,\pm 1) = 0, \tag{13}$$

solutions to the physical problem for (r,z) in $[0,\,1] \times [-1,\,1]$ are then obtained as follows:

$$X(rR,zd) = \chi(r,z), \qquad \mathcal{V}(rR,zd) = \frac{\hbar^2}{2\mu d^2}V(r,z), \qquad \mathcal{E} = \frac{\hbar^2}{2\mu d^2}E. \tag{14}$$

Solving Eq. (11), and then scaling the solutions using Eq. (14) gives the family of WEFs representing the wavefunction inside the scattering region [16,18].

In general, the scattering potential $V(r,z)$ is not separable. However, if the potential can be represented as $V(r,z) = U(r) + W(z)$, which arises in some cases of physical interest, Eq. (11) is separable [16]. Using separation of variables, $\chi(r,z) \to \phi(r)\varphi(z)$, and applying the boundary conditions, a complete basis for constructing the WEFs is obtained.

The complete orthonormal set of radial basis functions is expressed in terms of Fourier–Bessel functions [23,24]. The functions $\left\{\phi_k^{(m)}(r)\right\}_{k=1,2,\ldots}$ satisfy

$$T_r^{(m)}\phi_k^{(m)}(r) = {j_{m,k}}^2\phi_k^{(m)}(r), \tag{15}$$

where

$$\phi_k^{(m)}(r) = \frac{\sqrt{2}J_m\left(j_{m,k}r\right)}{J_{m+1}\left(j_{m,k}\right)}, \qquad \phi_k^{(m)}(1) = 0, \tag{16}$$

and $j_{m,k}$ is the kth zero of the Bessel function $J_m(x)$. The orthonormality condition reads

$$\int_0^1 \phi_K^{(m)}(r)\phi_k^{(m)}(r)rdr = \delta_{K,k}. \tag{17}$$

The axial basis functions $\{\varphi_p(z)\}_{p=0,1,\ldots}$ satisfy

$$-\varphi_p''(z) = \left(\frac{\pi p}{2}\right)^2 \varphi_p(z), \tag{18}$$

along with the boundary condition (14), $\varphi_p'(\pm 1) = 0$, and are given by Eq. (6). The general 2D WEF then reads

$$\chi(r,z) = \sum_{k=1}^{\infty}\sum_{p=0}^{\infty} c_{k,p}\phi_k^{(m)}(r)\varphi_p(z). \tag{19}$$

Introducing (in Dirac notation)

$$|\beta\rangle = \phi_k^{(m)}(r)\varphi_p(z), \tag{20}$$

where the label β corresponds to the set of indices k and p, then

$$\langle\alpha|\beta\rangle = \delta_{\alpha,\beta}, \tag{21}$$

due to the orthonormality of the radial and axial basis functions.

The matrix elements of the Hamiltonian in Eq. (11) yield the matrix eigenvalue problem,

$$\langle\alpha\,|H|\,\beta\rangle = \langle\alpha\,|T|\,\beta\rangle + \langle\alpha\,|V|\,\beta\rangle = E\,\langle\alpha|\beta\rangle\,. \tag{22}$$

With the above choice of radial and axial basis functions, $\langle\alpha\,|T|\,\beta\rangle$ is a diagonal matrix

$$\langle\alpha\,|T|\,\beta\rangle = \left(s^2 {j_{m,k}}^2 + \left(\frac{\pi p}{2}\right)^2\right)\delta_{\alpha,\beta}. \tag{23}$$

So, after computing the matrix elements of the potential, $\langle\alpha\,|V|\,\beta\rangle$, one obtains E_l and χ_l following the usual steps [5,6,16,18].

2.3. 2D matrix elements

Matrix elements of the potential $V(r,z)$ can be obtained by computing the radial integral

$$V_{k,K}^{(m)}(z) = \int_0^1 \phi_k^{(m)}(r)V(r,z)\phi_K^{(m)}(r)rdr, \tag{24}$$

followed by integration over z,

$$\langle\alpha\,|V(r,z)|\,\beta\rangle = V_{\alpha,\beta} = \int_{-1}^{1} \varphi_p(z)V_{k,K}^{(m)}(z)\varphi_P(z)dz. \tag{25}$$

This axial integral (25) can be computed efficiently using the DCT.

Now we focus on efficient computation of the radial integral. Instead of using the orthonormal Fourier–Bessel radial basis [23,24], one can use *any* complete basis. One simple choice, vanishing at $r = 1$, is the trigonometric basis

$$|k\rangle = u_k(r) := \sqrt{2}\cos\left((2k-1)\frac{\pi}{2}r\right), \tag{26}$$

where $k \geq 1$. This basis is complete and orthonormal, $\int_0^1 u_k(r)u_K(r)\,dr = \delta_{k,K}$. The kinetic energy operator matrix is no longer diagonal, but evaluation of the matrix elements of the potential is simplified. In the next section, efficient computation of radial integrals in this basis is performed, also using the DCT (of type 4: DCT-IV).

However, the trigonometric basis $u_k(r)$ is not an eigenfunction of the kinetic energy operator:

$$T_r^{(m)}u_k(r) = \left(\frac{m^2}{r^2} + \frac{\pi^2(2k-1)^2}{4}\right)u_k(r) + \frac{\pi(2k-1)}{\sqrt{2}r}\sin\left((2k-1)\frac{\pi}{2}r\right), \tag{27}$$

which means that the kinetic energy operator matrix is no longer diagonal. Because of the singularity of $T_r^{(m)}u_k(r)$ at $r = 0$, to obtain *finite* matrix elements, we consider $\langle k|r^2T_r^{(m)}|K\rangle$. The diagonal and off-diagonal entries read

$$\langle k\,|r^2T_r^{(m)}|\,K\rangle = \begin{cases} \dfrac{\pi^2(2k-1)^2}{12} + m^2 & k = K \\ \dfrac{(2k-1)(2K-1)(-1)^{k+K}(k(k-1)+3K(K-1)+1)}{2(k-K)^2(k+K-1)^2} & k \neq K. \end{cases} \tag{28}$$

The off-diagonal entries are *independent* of m so computation of the matrix elements of $\langle k|r^2T_r^{(m)}|K\rangle$ reduces to

$$\langle k\,|r^2T_r^{(m)}|\,K\rangle = \langle k\,|r^2T_r^{(0)}|\,K\rangle + m^2\delta_{k,K}. \tag{29}$$

Note that the matrix $\langle k|r^2 T_r^{(0)}|K\rangle$ is *not* symmetric under $k \leftrightarrow K$ as the difference

$$\langle k\,|r^2 T_r^{(0)}|\,K\rangle - \langle K\,|r^2 T_r^{(0)}|\,k\rangle = -\frac{(2k-1)(2K-1)(-1)^{k+K}}{(k-K)(k+K-1)} \tag{30}$$

is nonzero. However, computing the *generalized eigensystem* of $\langle k|r^2 T_r^{(m)}|K\rangle$ with respect to $\langle k\,|r^2|\,K\rangle$ is *equivalent* to using the Fourier–Bessel basis.

Introducing

$$|kp\rangle = u_k(r)\varphi_p(z), \tag{31}$$

we write the scaled 2D Schrödinger equation (11) as

$$s^2\langle kp\,|r^2 T_r^{(m)}|\,KP\rangle + \langle kp\,|r^2 T_z|\,KP\rangle + \langle kp\,|r^2 V(r,z)|\,KP\rangle = E\langle kp\,|r^2|\,KP\rangle. \tag{32}$$

Decomposing Eq. (32) into Kronecker products we have

$$s^2\langle k\,|r^2 T_r^{(m)}|\,K\rangle \otimes \langle p|P\rangle + \langle k\,|r^2|\,K\rangle \otimes \langle p\,|T_z|\,P\rangle + \langle kp\,|r^2 V(r,z)|\,KP\rangle = E\langle k\,|r^2|\,K\rangle \otimes \langle p|P\rangle\,, \tag{33}$$

where

$$\langle p|P\rangle = \delta_{p,P}, \qquad \langle p\,|T_z|\,P\rangle = \left(\frac{\pi p}{2}\right)^2 \delta_{p,P}. \tag{34}$$

The (modified) potential matrix element $\langle kp\,|r^2 V(r,z)|\,KP\rangle$ can be computed from the radial integral

$$v_{k,K}(z) = \int_0^1 u_k(r) r^2 V(r,z) u_K(r) dr, \tag{35}$$

followed by integration over z,

$$\langle kp\,|r^2 V(r,z)|\,KP\rangle = \int_{-1}^1 \varphi_p(z) v_{k,K}(z) \varphi_P(z) dz, \tag{36}$$

or *vice versa*.

3. Computation

3.1. Installation

To execute the code in this paper, the `WignerEisenbud` package needs to be loaded first. There are two ways to load the package:

3.1.1. Install the package

1. Start *Mathematica*. If you are already running *Mathematica*, to ensure that you start from a fresh kernel, quit the kernel using **Quit Kernel ▸ Local** under the **Evaluation** menu.
2. Open the Notebook version of this paper (`WignerEisenbudPaper.nb`).
3. Open `WignerEisenbud.m` in *Mathematica*.
4. Use the menu command **File ▸ Install**…to install `WignerEisenbud.m`.
5. Run `WignerEisenbud` by executing the following command:
   `<<WignerEisenbud‘`
6. To check that the `WignerEisenbud` package has been loaded, and to provide a list of functions defined in the package, execute the following command:
   `?*WignerEisenbud‘*`

3.1.2. Run the package

An alternative way, especially useful for testing, is to open `WignerEisenbud.m` within *Mathematica* and then push the **Run Package** button (in the top right corner). After that you can execute all commands in the `WignerEisenbudPaper.nb` notebook.

3.2. Matrix element computation using the DCT

The DCT is used for quadrature – essentially Clenshaw–Curtis quadrature and Fejér quadrature – and for solving partial differential equations by spectral methods [25–27]. There are four types of DCT, denoted DCT-I, II, III, IV. The boundary conditions determine which DCT applies. Higher dimensional DCTs result from the Kronecker product (equivalently, a composition) of one dimensional DCTs, and their computation is performed using a *row–column algorithm*. Free C libraries, such as *FFTW*, for computing fast DCTs (types I–IV) in one or more dimensions, of arbitrary size, are available [28]. In this paper, we use *Mathematica*'s `FourierDCT`, but it would be straightforward to modify our code to use external libraries.

DCT-II is the most commonly used transform, often simply referred to as "the DCT", and implemented in *Mathematica* as `FourierDCT` of type 2. To compute $\int_{-1}^1 \varphi_p(z) f(z) dz$ in the Wigner–Eisenbud basis (6), one uniformly samples the function $f(z_i)$ at n *evenly-spaced* points z_i, and then computes the DCT of the sampled data. Sampling of a 1D function f is performed by mapping f over the sample points:

```
SampledFunction[f_,n_,{a_,b_}] :=
  With[{d=1/n}, f /@ (a+(b-a)Range[d/2,1-d/2,d])]
```

Uniform sampling of f.

The set of trigonometric basis functions, $\left\{\varphi_p\left(z_i\right)\right\}_{p=1,2,\ldots,n}$, is computed directly using the *inverse* DCT, implemented in *Mathematica* by mapping the `FourierDCT` of type 3 over an $n \times n$ identity matrix:

```
AxialBasis[n_] := AxialBasis[n] =
  (z ↦ FourierDCT[z, 3]) /@ IdentityMatrix[n]
```

Construction of the basis using the inverse DCT.

To compute the matrix elements $\langle p\,|f|\,P\rangle$, one samples the function, $f_i = f\left(z_i\right)$, multiplies by the (transpose of the) sampled basis functions, $\left\{\varphi_p\left(z_i\right)\right\}_{p=1,2,\ldots,n}$, and then maps the (default type 2) DCT operation (using `FourierDCT`) over each row of the resulting matrix:

```
AxialMatrix[f_List] :=
  FourierDCT /@ Transpose[Transpose[AxialBasis[Length[f]]] f]
```

Computation of the matrix element $\langle p\,|f|\,P\rangle$ for a sampled function f.

The axial overlap matrix with elements $\langle p|P\rangle$ is just an $n \times n$ identity matrix:

```
AxialOverlapMatrix[n_] := AxialOverlapMatrix[n] = IdentityMatrix[n]
```

After defining the kinetic energy matrix (7):

```
AxialKineticMatrix[n_]:=AxialKineticMatrix[n]=
  π^2/4.0 DiagonalMatrix[Range[0,n-1]^2]
```

definition of the Hamiltonian matrix is straightforward:

```
HamiltonianMatrix[v_List]:=
  AxialKineticMatrix[Length[v]]+MatrixElements[v]
```

`WignerEisenbudSystem` returns sorted ordered pairs of eigenvalues and eigenvectors of the Hamiltonian matrix, computed using the *Mathematica* `Eigensystem` function:

```
EnergyOrdered[es_]:= Transpose[Sort[Transpose[Chop[es]]]]

WignerEisenbudSystem[v_List]:= WignerEisenbudSystem[v] =
  EnergyOrdered[Eigensystem[N[HamiltonianMatrix[v]]]]
```

3.3. 1D WEFs

After correcting the coefficients to take into account the DCT normalization, we then contract the eigenvectors with the basis functions (6), normalize and multiply the result by the overall factor $\sqrt{n/2}$ to yield (a discrete numerical approximation to) the set of normalized WEFs, $\{\chi_l(z)\}_{l=1,2,\ldots,n}$, implemented as `WignerEisenbudFunctions`:

```
WignerEisenbudFunctions[v_List]:=
Module[
  {n=Length[v],evecs=Last[WignerEisenbudSystem[v]]},
    evecs[[All,1]]=2evecs[[All,1]];
    Sqrt[n/2] Normalize /@ (evecs.AxialBasis[n])
]
```

Using *Mathematica*'s `Manipulate` functionality, it is straightforward to write a dynamic interface for displaying the WEF for square-well potentials of various widths and depths. A static version of the output of this interface is shown in Fig. 2. The dependence of the Wigner–Eisenbud energies and functions upon the width and depth of the potential well is qualitatively as expected. Fig. 3 displays even and odd degenerate levels for a double potential barrier.



Fig. 2. A dynamic interface for displaying low-order squared WEFs, $\chi_l(z)^2$, shifted upwards by their eigenvalue E_l, for square-well potentials of various widths and depths, in dimensionless units. A scale factor is applied to $\chi_l(z)^2$ to improve the visualization.



Fig. 3. Low level WEFs $\chi_l(z)$ shifted upwards by their eigenvalue E_l for a double potential barrier of height $V = 500$, barrier width $b = 0.25$, and center to center separation of 0.75, in dimensionless units.

3.4. 2D WEFs

The discrete basis in the radial direction is, again, constructed using the `FourierDCT`, this time of type 4 (which satisfies the boundary condition at $r = 1$, vanishing there):

```
RadialBasis[n_]:= RadialBasis[n] =
  (z↦FourierDCT[z, 4]) /@ IdentityMatrix[n]
```

Construction of the radial basis using the DCT.

The matrix elements $\langle k\,|f(r)|\,K\rangle = \int_0^1 u_k(r) f(r) u_K(r) dr$ are then computed for *uniformly* sampled $f(r)$:

```
RadialMatrix[f_List]:=
  (z↦FourierDCT[z, 4]) /@
    Transpose[Transpose[RadialBasis[Length[f]]] f]
```

Computation of the radial matrix elements $\langle k\,|f(r)|\,K\rangle$.

Construct the radial *kinetic* energy matrix elements, $\left\langle k\left|r^2 T_r^{(0)}\right|l\right\rangle$, as defined by Eq. (28) and, for general m, Eq. (29):

```
RadialKineticMatrix[n_]:=RadialKineticMatrix[n]=
SparseArray[
  {
    {k_,k_} → 1/12 π^2 (1-2 k)^2,
    {k_,l_} /; k ≠ l →
      ((2 k-1) (2 l-1) (-1)^(k+l) ((k-1) k+3 (l-1) l+1))/
        (2 (k-l)^2 (k+l-1)^2)
  },
  {n,n}
]
```

```
RadialKineticMatrix[m_,n_]:= RadialKineticMatrix[m,n] =
  RadialKineticMatrix[n] + m^2 IdentityMatrix[n]
```

Construct the $n \times n$ overlap matrix with elements $\left\langle k\left|r^2\right|K\right\rangle$, by sampling $r \mapsto r^2$:

```
RadialOverlapMatrix[n_]:= RadialOverlapMatrix[n] =
  RadialMatrix[SampledFunction[r↦r^2, n, {0, 1}]]
```

Sampling $f(r, z)$ uniformly $m \times n$ times over $[-1, 1] \times [0, 1]$ yields the matrix $f_{ij} = f\left(r_j, z_i\right)$:

```
SampledFunction[f_,{m_,n_}]:=
Module[
  {d=1/m,e=1/n},
    Table[f[r,z], {z,-1+d,1-d,2 d}, {r,e/2,1-e/2,e}]
]
```

Uniform sampling of f.

To construct the matrix elements $\langle kp\,|f(r, z)|\,KP\rangle$ one first maps `RadialMatrix` over the sampled function, $f_{i,j}$, resulting in a rank-3 tensor of radial integrals. The `AxialMatrix` function is then mapped at the appropriate level over the required transposition of this tensor to compute the axial integrals. Finally `ArrayFlatten` is used to generate the matrix:

```
MatrixElements[f_?MatrixQ,{p_,k_}]:=
Chop[ArrayFlatten[
  Map[AxialMatrix[#][[1;;p,1;;p]]&,
    Transpose[RadialMatrix /@ f, {3,2,1}][[1;;k,1;;k]], {2}]
  ]
]
```

Construction of the matrix with elements $\langle kp\,|f(r, z)|\,KP\rangle$.

This code was first tested on two classes of separable potentials: $V(r, z) = f(r) + g(z)$ and $V(r, z) = f(r) \times g(z)$, as the potential matrix elements in each case can be expressed as Kronecker products [29]:

$$\langle kp\,|f(r) + g(z)|\,KP\rangle = \langle k\,|f(r)|\,K\rangle \otimes \langle p|P\rangle + \langle k|K\rangle \otimes \langle p\,|g(z)|\,P\rangle\,, \tag{37}$$

$$\langle kp\,|f(r) \times g(z)|\,KP\rangle = \langle k\,|f(r)|\,K\rangle \otimes \langle p\,|g(z)|\,P\rangle\,. \tag{38}$$

Constructing the Hamiltonian matrix defined by Eq. (33) is now straightforward:

```
CircleTimes=KroneckerProduct;

HamiltonianMatrix[m_,v_?MatrixQ,{p_,k_},s_:1]:=
  s^2 RadialKineticMatrix[m,k]⊗AxialOverlapMatrix[p]+
  RadialOverlapMatrix[k]⊗AxialKineticMatrix[p]+
  MatrixElements[v SampledFunction[{r,z} ↦ r^2,Dimensions[v]],{p,k}]
```

`WignerEisenbudSystem` returns ordered pairs of eigenvalues and normalized eigenvectors of the Hamiltonian matrix, computed using *Mathematica*'s `Eigensystem` function with respect to $\left\langle k\left|r^2\right|K\right\rangle \otimes \langle p|P\rangle$ (see Eq. (33)):



Fig. 4. Nonseparable conical quantum dot potential in dimensionless units.

```
WignerEisenbudSystem[m_,v_?MatrixQ,{p_,k_},s_:1]:=
WignerEisenbudSystem[m,v,{p,k},s]=
  EnergyOrdered[
    Eigensystem[
      {HamiltonianMatrix[m,v,{p,k},s],
       RadialOverlapMatrix[k]⊗AxialOverlapMatrix[p]
      }
    ]
  ]
```

`WignerEisenbudEnergies` returns ordered eigenvalues of the Hamiltonian matrix:

```
WignerEisenbudEnergies[m_,v_?MatrixQ,{p_,k_},s_:1]:=
  First[WignerEisenbudSystem[m,v,{p,k},s]]
```

To apply the (integral) orthonormality condition,

$$\int_0^1 \int_{-1}^1 r\chi_L(r,z)\chi_l(r,z)dzdr = \delta_{L,l} \tag{39}$$

we define the inner product (using the built-in `AngleBracket` function):

```
⟨f_,g_⟩:=2 Total[SampledFunction[r ↦ r,Length[f],{0,1}].(f g)] /
  (Times @@ Dimensions[f]) /; Dimensions[f]==Dimensions[g]
```

Finally, we construct the set of normalized WEFs, $\{\chi_l(z)\}_{l=1,2,\ldots,n}$, implemented as `WignerEisenbudFunctions`, using the built-in `Normalize` function:

```
WignerEisenbudFunctions[m_,v_,{p_,k_},l_,s_:1]:=
Module[
  {evecs=Partition[WignerEisenbudSystem[m,v,{p,k},s][[2,1]],p]},
  Normalize[RadialBasis[k].evecs.AxialBasis[p],f ↦ Sqrt[⟨f,f⟩]]
]
```

Fig. 4 shows the nonseparable potential of a conical quantum dot, related to that considered by Racec et al. [19], and Fig. 5 displays some WEF solutions.

All previous example potentials were entered using a fixed piecewise function or symbolic relation. Supplementary code has been developed to allow users to input arbitrary axial potentials (see Fig. 6). This code interpolates between control points, using `Locator` objects, and computes the potential as an interpolating function, which can then be applied in the main code.

Fig. 7 shows a more realistic asymmetrical and irregular potential, which is common in the heterostructures semiconductor field [30,31], as we have many semiconductor layers and each layer has its own potential depth. Figs. 7 and 8 shows the behavior of the WEFs for these potentials in 1D and 2D respectively.

4. Conclusions

We present *Mathematica* code that uses the DCT for computing WEFs for arbitrary potentials in one dimension and two dimensions in cylindrical coordinates. Applications of this code to study the electrical transport characteristics of nanowires is presented in Raffah and Abbott [22]. Hybrid-symbolic methods are used to convert the 2D Schrödinger equation in cylindrical coordinates to a generalized matrix eigenproblem. All required (radial and axial) matrix elements are computed using the DCT, avoiding the need to compute integrals involving radial Fourier–Bessel basis functions. Using the DCT permits rapid computation of matrix elements, inheriting the numerical advantages of this transform [25–27]. It would be straightforward to modify our code to use fast external libraries, such as *FFTW* [28].



Fig. 5. Lowest-order 2D WEFs for the conical quantum dot potential with $m = 1$.



Fig. 6. Interface for dynamic potential construction. Smooth potentials are constructed by increasing the interpolation order. The number of `Locator` objects is controlled using ALT click.



Fig. 7. An irregular potential, typical of a semiconductor heterostructure.



Fig. 8. Lowest-order 2D WEFs for a heterostructure potential (Fig. 7, with no radial variation) and magnetic quantum number $m = 0$.

Acknowledgments

BR would like to acknowledge support of the King Abdulaziz University and thanks Dr. André B. Fletcher (UWA, School of Physics) for comments on a draft of this paper.

References

[1] E.P. Wigner, L. Eisenbud, Phys. Rev. 72 (1) (1947) 29. http://dx.doi.org/10.1103/PhysRev.72.29.
[2] P.G. Burke, I. Mackey, I. Shimamura, J. Phys. B 10 (1977) 12.
[3] P. Burke, K. Berrington, Atomic and Molecular Processes: An *R*-Matrix Approach, Institute of Physics Publishing, London, 1993.
[4] U. Wulf, J. Kučera, P.N. Racec, E. Sigmund, Phys. Rev. B 58 (1998) 16209.
[5] P.N. Racec, Transport phenomena and capacitance of open quantum semiconductor nanostructures, Ph.D. Thesis, Brandenburgische Technische Universität, 2002. URL: http://deposit.ddb.de/cgi-bin/dokserv?idn=965463613.
[6] E.R. Racec, Electrons and optical phonons in mesoscopic semiconductor heterostructures, Ph.D. Thesis, Brandenburgische Technische Universität, 2002. URL: http://deposit.ddb.de/cgi-bin/dokserv?idn=965466566.
[7] G.A. Nemnes, U. Wulf, P.N. Racec, J. Appl. Phys. 96 (2004) 596.
[8] J. Behrndt, H. Neidhardt, E.R. Racec, P.N. Racec, U. Wulf, J. Differential Equations 244 (10) (2008) 2545.
[9] M. Büttiker, Y. Imry, R. Landauer, S. Pinhas, Phys. Rev. B 31 (1985) 6207.
[10] J. Stein, H.-J. Stöckmann, U. Stoffregen, Phys. Rev. Lett. 75 (1995) 53.
[11] H.U. Baranger, A.D. Stone, Phys. Rev. Lett. 63 (1989) 414.
[12] J. Masek, P. Lipavsky, B. Kramer, J. Phys.: Condens. Matter 1 (1989) 6395.
[13] S. He, S.D. Sarma, Phys. Rev. B 40 (1989) 3379.
[14] A. Szafer, A.D. Stone, Phys. Rev. Lett. 62 (1989) 300.
[15] G. Kirczenow, J. Phys.: Condens. Matter 1 (1989) 305.
[16] P.N. Racec, E.R. Racec, H. Neidhardt, Phys. Rev. B 79 (2009) 155305.
[17] P.N. Racec, E.R. Racec, H. Neidhardt, *R*-matrix formalism for electron scattering in two dimensions, Weiers.-Inst. Angew. Analy. Stoch. (ISSN: 0946-8633) (2009).
[18] H. Neidhardt, P.N. Racec, Modeling of quantum transport in nanowires, Berlin, 2008, ISSN: 1437-7489.
[19] P.N. Racec, E.R. Racec, H. Neidhardt, *R*-matrix formalism for electron scattering in two dimensions with applications to nanostructures with quantum dots, in: A. Aldea, V. Barsan (Eds.), Trends in Nanophysics: Theory, Experiment and Technology, Springer, 2010.
[20] T. Wu, T. Ohmura, Quantum Theory of Scattering, Prentice-Hall INC., 1962.
[21] E.R. Racec, U. Wulf, P.N. Racec, Phys. Rev. B 82 (2010) 085313.
[22] B.M. Raffah, P.C. Abbott, J. Comput. Theor. Nanosci. (2013) (in press).
[23] Digital Library of Mathematical Functions. Release 1.0.5, 2012-10-01. URL: http://dlmf.nist.gov/10.23#P6.
[24] R.K. Jain, S.R.K. Iyengar, Advanced Engineering Mathematics, Narosa Publishing House, 2002.
[25] C.W. Clenshaw, A.R. Curtis, Numer. Math. 2 (1960) 197.
[26] K.B. Howell, Principles of Fourier Analysis, Chapman and Hall/CRC, 2001.
[27] L.N. Trefethen, SIAM Rev. 50 (1) (2008) 67–87.
[28] M. Frigo, S.G. Johnson, The design and implementation of FFTW3, Proc. IEEE 93 (2005) 216–231. URL: http://www.fftw.org/.
[29] J.W. Brewer, IEEE Trans. Circuits Syst. 25 (9) (1978).
[30] C. Buizert, F.H.L. Koppens, M. Pioro-Ladrière, H. Tranitz, I.T. Vink, S. Tarucha, W. Wegscheider, L.M.K. Vandersypen, Phys. Rev. Lett. 101 (2008) 226603.
[31] J.D. Cressler, Silicon Heterostructure Devices, Taylor & Francis Group, LLC, 2008.